Mathematica Project 3 Help

Lines and Planes (also see examples in our class lab notebooks)

In *Mathematica*, one can plot a parametrically defined line in space by using the function *Parametric-Plot3D:*

```
ParametricPlot3D[{2 + t, 3 - 4 t, t}, {t, -10, 10}, PlotRange → 20, AxesLabel → {x, y, z}]
```



A plane in space can be given by a simplified component equation Ax+By+Cz=D using *ContourPlot3D*:

```
ContourPlot3D[2 x - 3 y + z - 6 == 0, {x, -10, 10}, {y, -10, 10}, {z, -10, 10},
 Mesh → None, ContourStyle → Opacity[0.5], AxesLabel → {x, y, z}]
```



In the next example, we have two intersecting planes and the line of intersection plotted. (Command *Show* is used to combine the graphs in one.)
Additionally, the normal vectors to the planes and a vector the line is parallel to are shown:

```
Show[
 ContourPlot3D[{x + y + z == 1, x - 2 y + 3 z == 1},
  {x, -10, 10}, {y, -10, 10}, {z, -10, 10}, AxesLabel → {x, y, z},
  Mesh → None, ContourStyle -> Opacity[.5]],
 ParametricPlot3D[{1 + 5 t, -2 t, -3 t},
  {t, -10, 10}, PlotRange → 20, AxesLabel → {x, y, z}],
 Graphics3D[{Arrowheads[Small], Arrow[{{1, 0, 0}, {2, 1, 1}}],
   Arrow[{{1, 0, 0}, {2, -2, 3}}], Arrow[{{1, 0, 0}, {6, -2, -3}}]}]
]
```



Displaying a tangent line to a 3D curve for a given value of t

```
r1[t_] := {2 Cos[t], 3 Sin[t], t}
TangentLine[t_] := r1[2 Pi / 3] + t * r1'[2 Pi / 3]
(*vector equation of the tangent line*)
Show[
 (*tangent line*)
 ParametricPlot3D[{r1[t], TangentLine[t]}, {t, -2 Pi, 2 Pi}, AxesLabel → {x, y, z}],
 (*tangent vector*) Graphics3D[
  {Arrowheads[.03], Arrow[{r1[2 Pi / 3], r1[2 Pi / 3] + r1'[2 Pi / 3]}]}]
]
```



Using a slider to show the TNB-frame for a space curve:

```
r2[t_] := {Cos[t], Sin[t], t}(* our curve is a helix *)
rT[t_] := r2'[t] / Sqrt[r2'[t].r2'[t]]
(* computing T, the unit tangent to the curve *)
rn[t_] := rT'[t] / Sqrt[r2'[t].r2'[t]](* dT/ds,
a normal to the curve -- perpendicular to the unit tangent vector *)
rN[t_] := rn[t] / Sqrt[rn[t].rn[t]](* N, the principal unit normal vector *)
rB[t_] := Cross[rT[t], rN[t]] (* B, the unit binormal vector *)


Manipulate[
 Show[
  ParametricPlot3D[r2[t], {t, 0, 4 Pi}, PlotRange -> {{-2, 2}, {-2, 2}, { 0,  4 Pi}}],
  Graphics3D[{Arrowheads[.025],
             Arrow[{r2[ts], r2[ts] + rT[ts]}],
    Text[Style["T", Bold], {r2[ts] + rT[ts]}],
             Red, Arrow[{r2[ts], r2[ts] + rN[ts]}],
    Text[Style["N", Bold], {r2[ts] + rN[ts]}],
             Blue, Arrow[{r2[ts], r2[ts] + rB[ts]}],
    Text[Style["B", Bold], {r2[ts] + rB[ts]}]
   }]
 ],
 {ts, 0, 4 Pi}
]
```


ts
7.98374
−2
−1
0
1
2
2
1
0
−1
−2
10
5
0
T
B
N